

Admin

Check in: Lab0, first week

Assign0 run-through git workflow
due before Tue lab

OH schedule being settled
See calendar on course home page




Today: Let there be light!

More on RISC-V assembly, instruction encoding

Peripheral access through memory-mapped registers


Goal: blink an LED

Branch instructions for control flow



again:
addi a0,a0,-1
bne a0,zero,again


Default is straight line execution $PC = PC + 4$



Jump/branch changes PC to target


Load and store instructions

lw a2,0x40(a0)



offset expressed as immediate
add offset to base to compute memory address
read/write contents at that address

sw a1,0x30(a0)




MangoPi memory is little-endian


most-significant-byte (MSB)



least-significant-byte (LSB)



little-endian
(LSB first)



The 'little-endian' and 'big-endian' terminology which is used to denote the two approaches [to addressing memory] is derived from Swift's *Gulliver's Travels*. The inhabitants of Lilliput, who are well known for being rather small, are, in addition, constrained by law to break their eggs only at the little end. When this law is imposed, those of their fellow citizens who prefer to break their eggs at the big end take exception to the new rule and civil war breaks out. The big-endians eventually take refuge on a nearby island, which is the kingdom of Blefuscus. The civil war results in many casualties.

Read: Holy Wars and a Plea For Peace, D. Cohen

Let's write a program together!

A certain memory address holds the input number.
Count the "on" bits in that number and store count at
neighboring address

```
// read value from memory location 0x10000000
// loop to count "on" bits in value
// write final count to memory location 0x10000003
```



Ripes visual simulator

The screenshot shows the Ripes visual simulator interface. On the left, there's a sidebar with icons for Processor, Cache, Memory, and I/O. The main area has tabs for Source code, Input type (Assembly selected), C Executable code, View mode (Binary and Disassembled selected), and a zoom tool. The Source code tab shows assembly code:

```
3 .text
4
5 lui a3, 0x10000
6 mv a1, zero
7 lw a0, 0(a3)
8
9 loop:
10 andi a2,a0,1
11 add a1,a1,a2
12 srl a0,a0,1
13 bne a0,zero,loop
14
15 sw a1, 4(a3)
```

The Disassembled tab shows the corresponding binary instructions:

Address	Word	Byte 0	Byte 1	Byte 2	Byte 3
0x10000010	X	X	X	X	X
0x1000000c	X	X	X	X	X
0x10000008	X	X	X	X	X
0x10000004	0x00000005	0x05	0x00	0x00	0x00
0x10000000	0x00001234	0x34	0x12	0x00	0x00
0x0fffffc	X	X	X	X	X
0x0fffff8	X	X	X	X	X
0x0fffff4	X	X	X	X	X
0x0fffff0	X	X	X	X	X

To the right is a register table:

Name	Alias	Value
x0	zero	0x00000000
x1	ra	0x00000000
x2	sp	0x7fffffff0
x3	gp	0x10000000
x4	tp	0x00000000
x5	t0	0x00000000
x6	t1	0x00000000
x7	t2	0x00000000
x8	s0	0x00000000
x9	s1	0x00000000
x10	a0	0x00000000
x11	a1	0x00000005
x12	a2	0x00000001
x13	a3	0x10000000
x14	a4	0x00000000
x15	a5	0x00000000
x16	a6	0x00000000
x17	a7	0x00000000
x18	s2	0x00000000
x19	s3	0x00000000
x20	s4	0x00000000
x21	s5	0x00000000

At the bottom, there are buttons for Display type (Hex), Go to register, Go to section, and Processor: Single-cycle processor ISA: RV32IM.

<https://ripes.me>

Instruction encoding

Another way to understand the design of an ISA is to look at how the bits are used in the instruction encoding.

In RISC-V, each instruction is encoded in **32 bits***

Instructions are organized into six groups, some features are common within/across groups.

The encoding design intentionally reuses same bits for common features to simplify decode circuitry.

*Compressed extension adds 16-bit compact encoding for some insns

RISC-V instruction encoding

32-bit instruction format

	31	30	29	28	27	26	25	24	23	22	21	20	19	18	17	16	15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0	
R																																	
	func		rs2		rs1		func		rd		opcode																						
I									rs1		func		rd		opcode																		
SB										rs1		func		immediate		opcode																	
UJ																					rd		opcode										

6 instruction types (R, I, S/B, U/J)

Regularity in bit placement to ease decoding

Sparse instruction encoding (room for growth)

ALU R-type encoding


31	27	26	25	24	20	19	15	14	12	11	7	6	0	
funct7		rs2		rs1		funct3		rd		opcode		R-type		
imm[11:0]				rs1		funct3		rd		opcode		I-type		
imm[11:5]		rs2		rs1		funct3		imm[4:0]		opcode		S-type		
imm[12 10:5]		rs2		rs1		funct3		imm[4:1 11]		opcode		B-type		
		imm[31:12]						rd		opcode		U-type		

add **x3**, **x1**, **x2**

0000000	rs2	rs1	000	rd	0110011	ADD
0100000	rs2	rs1	000	rd	0110011	SUB
0000000	rs2	rs1	001	rd	0110011	SLL
0000000	rs2	rs1	010	rd	0110011	SLT
0000000	rs2	rs1	011	rd	0110011	SLTU
0000000	rs2	rs1	100	rd	0110011	XOR
0000000	rs2	rs1	101	rd	0110011	SRL
0100000	rs2	rs1	101	rd	0110011	SRA
0000000	rs2	rs1	110	rd	0110011	OR
0000000	rs2	rs1	111	rd	0110011	AND

00000000**0001000001**000**0001**10110011
 0 0 2 0 8 1 B 3

ALU I-type encoding



imm[11:0]	rs1	000	rd	0010011	ADDI
imm[11:0]	rs1	010	rd	0010011	SLTI
imm[11:0]	rs1	011	rd	0010011	SLTIU
imm[11:0]	rs1	100	rd	0010011	XORI
imm[11:0]	rs1	110	rd	0010011	ORI
imm[11:0]	rs1	111	rd	0010011	ANDI

Your turn!

xori a0,zero,21

000000010101000000100010100010011
0 1 5 0 4 5 1 3

Know your tools: assembler

The *assembler* reads assembly instructions (*text*) and outputs as machine-code (*binary*). The reverse process is called *disassembly*.

This translation is mechanical, fully deterministic.

```
$ riscv64-unknown-elf-as add.s -o add.o  
  
$ ls -l add.o  
928 add.o  
  
$ riscv64-unknown-elf-objcopy add.o add.bin -O binary  
  
$ ls -l add.bin  
4 add.bin  
  
$ hexdump -C add.bin  
00000000  b3 81 20 00
```

Let there be light






Computer *peripheral* interfaces to outside world

GPIO pins are peripherals

Let's learn how to control a GPIO pin with code!

Mango Pi GPIO



9.7 GPIO


9.7.1 Overview

The general purpose input/output (GPIO) is one of the blocks controlling the chip multiplexing pins. The D1-H supports 6 groups of GPIO pins. Each pin can be configured as input or output and these pins are used to generate input signals or output signals for special purposes.

The Port Controller has the following features:

- 6 groups of ports (PB, PC, PD, PE, PF, PG)
- Software control for each signal pin
- Data input (capture)/output (drive)
- Each GPIO peripheral can produce an interrupt


Connect LED to GPIO PB0



Set/clear GPIO will change output voltage
set 1 -> 3.3V
clear 0 -> 0.0V



Complete circuit
(to ground pin)




Q: What is purpose of resistor in this circuit?

Memory Map

Peripheral registers are mapped into address space

Read/write to memory address controls peripheral


This is
"memory-mapped IO"
(MMIO)



The screenshot shows a table from the Allwinner DI-H User Manual titled "3 System" under "3.1 Memory Mapping". The table lists memory modules, their addresses, and sizes. The "SP0 (SYS Domain)" section includes GPIO, PWM, CCU, IR_TX, LEDC, GPADC, LRADC, THS, TPADC, IOMMU, Audio Codec, DMIC, I2S0, I2S1, I2S2, CWA, and TIMER. The "SP1 (SYS Domain)" section includes UART0 and UUART1. The "GPIO" row is highlighted with a yellow background.

Module	Address (It is for Cluster CPU)	Size
SP0 (SYS Domain)		
GPIO	0x0200 0000 – 0x0200 07FF	2 KB
PWM	0x0200 0C00 – 0x0200 OFFF	1 KB
CCU	0x0200 1000 – 0x0200 1FFF	4 KB
IR_TX	0x0200 3000 – 0x0200 33FF	1 KB
LEDC	0x0200 8000 – 0x0200 83FF	1 KB
GPADC	0x0200 9000 – 0x0200 93FF	1 KB
LRADC	0x0200 9800 – 0x0200 9BFF	1 KB
THS	0x0200 9400 – 0x0200 97FF	1 KB
TPADC	0x0200 9C00 – 0x0200 9FFF	1 KB
IOMMU	0x0201 0000 – 0x0201 FFFF	64 KB
Audio Codec	0x0203 0000 – 0x0203 0FFF	4 KB
DMIC	0x0203 1000 – 0x0203 13FF	1 KB
I2S0	0x0203 2000 – 0x0203 2FFF	4 KB
I2S1	0x0203 3000 – 0x0203 3FFF	4 KB
I2S2	0x0203 4000 – 0x0203 4FFF	4 KB
CWA	0x0203 6000 – 0x0203 63FF	1 KB
TIMER	0x0205 0000 – 0x0205 OFFF	4 KB
SP1 (SYS Domain)		
UART0	0x0250 0000 – 0x0250 03FF	1 KB
UUART1	0x0250 0400 – 0x0250 07FF	1 KB

Ref: **DI-H User Manual p.45**



9.7.4 Register List

Module Name	Base Address
GPIO	0x02000000

Register Name	Offset	Description
PB_CFG0	0x0030	PB Configure Register 0
PB_CFG1	0x0034	PB Configure Register 1
PB_DAT	0x0040	PB Data Register
PB_DRV0	0x0044	PB Multi_Driving Register 0
PB_DRV1	0x0048	PB Multi_Driving Register 1
PB_PULL0	0x0054	PB Pull Register 0
PC_CFG0	0x0060	PC Configure Register 0
PC_DAT	0x0070	PC Data Register
PC_DRV0	0x0074	PC Multi_Driving Register 0
PC_PULL0	0x0084	PC Pull Register 0

Access "Configure Register" to select pin function

Access "Data Register" to change pin value

9.7.3.2 GPIO Multiplex Function

Table 9-21 to Table 9-26 show the multiplex function pins of the D1-H.



NOTE


For each GPIO, Function0 is input function; Function1 is output function; Function9 to Function13 are reserved.

Table 9-21 PB Multiplex Function

GPIO Port	Function 2	Function 3	Function 4	Function 5	Function 6	Function 7	Function 8	Function 14
PB0	PWM3	IR-TX	TWI2-SCK	SPI1-WP/DBI-TE	UART0-TX	UART2-TX	OWA-OUT	PB-EINT0
PB1	PWM4	I2S2-DOUT3	TWI2-SDA	I2S2-DIN3	UART0-RX	UART2-RX	IR-RX	PB-EINT1
PB2	LCD0-D0	I2S2-DOUT2	TWI0-SDA	I2S2-DIN2	LCD0-D18	UART4-TX		PB-EINT2
PB3	LCD0-D1	I2S2-DOUT1	TWI0-SCK	I2S2-DIN0	LCD0-D19	UART4-RX		PB-EINT3
PB4	LCD0-D8	I2S2-DOUT0	TWI1-SCK	I2S2-DIN1	LCD0-D20	UART5-TX		PB-EINT4

GPIO Configure Register

PB Config0 @0x02000030



4 bits per GPIO pin

32-bit register stores config for 8 pins

Pin select is 4 bits (16 options)

0:input, 1:output


2-8: alt fns, 9-13:reserved

14:interrupt, 15:disabled

Offset: 0x0030			Register Name: PB_CFG0
Bit	Read/Write	Default/Hex	Description
3:0	R/W	0xF	PB0_SELECT PB0 Select 0000:Input 0001:Output 0010:PWM3 0011:IR-TX 0100:TWI2-SCK 0101:SPI1-WP/DBI-TE 0110:UART0-TX 0111:UART2-TX 1000:OWA-OUT 1001:Reserved 1110:PB-EINT0 1111:IO Disable

Ref: [DI-H User Manual p.1097](#)

GPIO Data Register



I bit per GPIO pin

Value is 1 if high, 0 low

9.7.5.3 0x0040 PB Data Register (Default Value: 0x0000_0000)

Offset: 0x0040			Register Name: PB_DAT
Bit	Read/Write	Default/Hex	Description
31:13	/	/	/
12:0	R/W	0x0	PB_DAT If the port is configured as the input function, the corresponding bit is the pin state. If the port is configured as the output function, the pin state is the same as the corresponding bit. The read bit value is the value set up by software. If the port is configured as a functional pin, the undefined value will be read.

Ref: DI-H User Manual p.1098

Using xfel

BOOT ROM of Mango Pi starts in "FEL" by default
(Firmware Exchange Loader)

FEL listens on USB port

Connect laptop to Mango Pi USB OTG port

Run `xfel` on your laptop to communicate with FEL on Pi

Can use `xfel` to peek and poke memory addresses, load and execute programs, and more

```
$ xfel write32 0x02000030 0x1
$ xfel write32 0x02000040 0x1
```

on.s

```
lui    a0,0x2000  
addi   a1,zero,1  
sw     a1,0x30(a0) }  
  
sw     a1,0x40(a0) }  
  
loop:  
      j  loop } }
```

Select output fn for pin PB0

Set pin PB0 data value to 1

Loop infinitely

Reminders:

PB CFG0 register @ 0x02000030

PB DATA register @ 0x02000040

Build and execute

```
$ riscv64-unknown-elf-as on -o on  
  
$ riscv64-unknown-elf-objcopy on.o on.bin -O binary  
  
$ mango-run on.bin  
    xfel ddr d1  
    xfel write 0x40000000 on.bin  
    xfel exec 0x40000000
```

- 1) assembler translates assembly insns to machine encoding
- 2) objcopy extracts raw binary
- 3) mango-run: xfel init dram, load program, execute it

blink.s

```
lui      a0,0x2000
addi    a1,zero,1
sw      a1,0x30(a0)      # config PB0 as output

loop:
xori    a1,a1,1          # xor ^ 1 invert bit 0
sw      a1,0x40(a0)      # flip bit on<->off

        lui    a2,0x3f00      # busy loop wait
delay:
addi    a2,a2,-1
bne    a2,zero,delay

j      loop               # repeat forever
```

Key concepts so far

Translation between assembly and machine encoding
(assembler/disassembler)

Mango Pi uses little-endian memory organization

General purpose IO (GPIO), peripheral registers, MMIO

Tools to build, load, execute a program

Resources to keep handy

RISC-V one-page guide

Ripes simulator

Mango Pi pinout

DI-H User Manual